Tema 2
Cálculos básicos con wxMaxima

Objetivos

1. Llevar a cabo cálculos básicos con variables.
2. Llevar a cabo cálculos básicos con expresiones algebraicas.
3. Llevar a cabo cálculos básicos con polinomios y fracciones racionales.
4. Resolución básica de ecuaciones e inecuaciones.
5. Funciones elementales: definición y sintaxis con wxMaxima.

Contenidos

02-1. Cálculos básicos con variables.
02-2. Expresiones algebraicas.
02-3. Polinomios y fracciones racionales
02-4. Resolución básica de ecuaciones.
02-5. Resolución básica de sistemas de ecuaciones.
02-6. Resolución básica de inecuaciones.
02-7. Funciones elementales con wxMaxima.

Referencies

AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.

G07 GLASNER, MOSES (2007)
A Maxima Guide for Calculus Students

JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.

RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima

RG07 RODRÍGUEZ GALVÁN, J. RAFAEL (2007)
Maxima con wxMaxima: software libre en el aula de matemáticas

RR08 RODRÍGUEZ RIOTORTO, MARIO (2008)
Primeros pasos en Maxima

RV09 RODRÍGUEZ, MARIO; VILLATE, JAIME (2009)
Manual de Maxima.

VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

02-1.- Cálculos básicos con variables

Los cálculos de este apartado se desarrollan en el fichero **Tema_02-1.wxm**.

A menudo los cálculos matemáticos se estructuran y se llevan a cabo con variables, es decir, símbolos que pueden tomar valores numéricos diferentes. Para designar una variable se puede usar una letra, un conjunto de letras o un conjunto de letra y número. Son ejemplos de variables: x, y, z, xx, yy, x1, x2, y1, y2, etc.

La sintaxis de la asignación de variables con wxMaxima es sencilla: se trata de escribir el símbolo o símbolos que se quiere aplicar a la variable, a continuación el signo ":" y finalmente el valor numérico asignado. Si se pone al final el signo "$" se asignará el valor a la variable pero no se mostrará y si al final se pone el signo ";" se asignará y se mostrará. Veamos unos ejemplos de variables y de asignación de valores numéricos a variables, con la realización de unos cálculos elementales:

```
(%i1) a1:115.654$          b1:401.928$
(%i3) a:234.34567;          b:567.4321;
(%o3) 234.34567
(%o4) 567.4321
(%i5) a+b;          a*b;          2.145*a-34.45*b;
(%o5) 801.77777
(%o6) 132975.255654007
(%o7) -19045.36438285
```

Sintaxis para mostrar el símbolo de una variable además de su valor numérico, asignado o calculado:

```
(%i8) '(a+b)=a+b;        '(ab)=a*b;
      '(2.145*a-34.45*b)=2.145*a-34.45*b;
(%o8) b+a=801.77777
(%o9) ab=132975.255654007
(%o10) 2.145 a-34.45 b=-19045.36438285
```

Ejemplo de cómo se pueden considerar o definir nuevas variables, asignar valores numéricos a estas variables y calcular el valor de una expresión en la que intervienen estas variables nuevas y variables anteriores:

```
(%i11) x:987.65$       y:432.19$
       45.34*a+24.345*b-67.765*x^2+34.21*sqrt(y);
(%o13) -6.6076389622216567 10^7
```

Ilustración de cómo se pueden definir variables nuevas a partir de otras variables definidas con anterioridad y como se pueden hacer cálculos con estas variables:

```
(%i14) R1:x^2+y^2;          R2:x+y;       R1+R2;
(%o14) 1162240.7186
  (%o15) 1419.84
  (%o16) 1163660.5586
(%i17) '(R1)=R1:x^2+y^2;      '(R2)=R2:x+y;
         '(R1+R2)=R1+R2;
(%o17) R1 = 1162240.7186
  (%o18) R2 = 1419.84
  (%o19) R2 + R1 = 1163660.5586
```

Ejemplo de la aplicación de las variables en un procedimiento para el cálculo del área de un círculo:

```
(%i20) Radi:23.456;       '(Area_cercle)=%pi*Radi^2, numer;
(%o20) 23.456
  (%o21) Area_cercle = 1728.453811460717
```

Ejemplo de la aplicación de las variables en un procedimiento para el cálculo del área de un triángulo:

```
(%i22) Base:123.45;       Altura:6.7;
       '(Area_triangle)=Base*Altura/2;
(%o22) 123.45
  (%o23) 6.7
  (%o24) Area_triangle = 413.5575
```

Ejemplo de la aplicación de las variables en un procedimiento para el cálculo del área de un trapecio:

```
(%i25) B1:12.34;          B2:7.543;
       H:2.345;           '(Area_Trapezi)=(B1+B2)*H/2;
(%o25) 12.34
  (%o26) 7.543
  (%o27) 2.345
  (%o28) Area_Trapezi = 23.3128175
```

Puede interesar modificar el valor asignado a una variable o a varias variables. Con esta finalidad hay que aplicar la instrucción “kill(all)$” para anular (“matar”) los valores asignados o calculados de las variables definidas o consideradas.

A partir de la ejecución de esta instrucción las variables no tienen ningún valor numérico asignado hasta que se haga una nueva asignación de forma específica. Si se elaboran diferentes cálculos con una cierta variable, se recomienda anular su valor antes de iniciar un nuevo procedimiento. Así, por ejemplo en relación a las variables aplicadas en cálculos anteriores se tendría:

```
(%i29) kill(all)$
       a+b;     a*b;
       R1;      R2;
       Radi;    Base;
(%o1) b+a
(%o2) a b
(%o3) R1
(%o4) R2
(%o5) Radi
(%o6) Base
```

02-2.- Expresiones algebraicas

En las sesiones anteriores se ha visto como se llevan a cabo cálculos aritméticos elementales y cálculos con variables. Se trata ahora de ver cómo el programa es capaz de trabajar con expresiones algebraicas.

Los cálculos de este apartado se desarrollan en el fichero **Tema_02-2.wxm**.

Ejemplo de expresión algebraica:

```
(%i1) a^2 + b/a + b;
```

(%o1) $\frac{b}{a}+b+a^2$

Cuando se quiere aplicar una instrucción al resultado inmediatamente anterior se puede hacer mediante la sintaxis instrucción(%), donde “instrucción” indica la función a realizar. Así, por ejemplo, si se quiere simplificar el resultado anterior y se quiere hacer con la instrucción “ratsimp” (simplificar):

```
(%i2) ratsimp(%);
```

(%o2) $\frac{(a+1)\,b+a^3}{a}$

De la misma manera, se puede expandir una fracción algebraica, que no es ni más ni menos que lo inverso del procedimiento anterior:

```
(%i3) expand(%);
```

(%o3) $\frac{b}{a}+b+a^2$

Ejemplos de potencias de binomios:

```
(%i4) (a+b)^2;
```

(%o4) $(b+a)^2$

```
(%i5) expand(%);
```

(%o5) $b^2+2\,a\,b+a^2$

```
(%i6) (a+b)^4;
```

(%o6) $(b+a)^4$

```
(%i7) expand(%);
```

(%o7) $b^4+4\,a\,b^3+6\,a^2\,b^2+4\,a^3\,b+a^4$

```
(%i8) (a+b)^6;
```

(%o8) $(b+a)^6$

(%i9) expand(%);

(%o9) $b^6+6\,a\,b^5+15\,a^2\,b^4+20\,a^3\,b^3+15\,a^4\,b^2+6\,a^5\,b+a^6$

Ejemplos de operaciones con expresiones algebraicas:

(%i10) (a+b)^6 -8*(a+b)^4 + 10;

(%o10) $(b+a)^6-8\,(b+a)^4+10$

(%i11) expand(%);

(%o11) $b^6+6\,a\,b^5+15\,a^2\,b^4-8\,b^4+20\,a^3\,b^3-32\,a\,b^3+15\,a^4\,b^2-48\,a^2\,b^2+6\,a^5\,b-32\,a^3\,b+a^6-8\,a^4+10$

(%i12) b^4+4*a*b^3+6*a^2*b^2-2*b^2+4*a^3*b-4*a*b+a^4-2*a^2+1;

(%o12) $b^4+4\,a\,b^3+6\,a^2\,b^2-2\,b^2+4\,a^3\,b-4\,a\,b+a^4-2\,a^2+1$

(%i13) factor(%);

(%o13) $(b+a-1)^2\,(b+a+1)^2$

(%i14) expand((a+1)*(b+2)^2-3*(a+b)^6);

(%o14) $-3\,b^6-18\,a\,b^5-45\,a^2\,b^4-60\,a^3\,b^3-45\,a^4\,b^2+a\,b^2+b^2-18\,a^5\,b+4\,a\,b+4\,b-3\,a^6+4\,a+4$

02-3.- Polinomios y fracciones racionales

Los cálculos de este apartado se desarrollan en el fichero **Tema_02-3.wxm**.

A menudo los cálculos con polinomios conllevan tener que aplicarlos más de una vez con un mismo polinomio; esto aconseja usar una asignación en cada uno de los polinomios que se aplican en los cálculos. La notación vista para las variables permite esta asignación.

Los polinomios son combinaciones lineales de las potencias de una variable. Se pueden definir con coeficientes expresados como parámetros generales. Así, por ejemplo:

```
(%i1) P1:a0+a1*x;
      P2:b0+b1*x+b2*x^2;
      P3:c0+c1*x+c2*x^2+c3*x^3;
```

(%o1) $a1\ x+a0$

(%o2) $b2\ x^2+b1\ x+b0$

(%o3) $c3\ x^3+c2\ x^2+c1\ x+c0$

```
(%i4) P1 = P1:a0+a1*x;
      P2 = P2:b0+b1*x+b2*x^2;
      P3 = P3:c0+c1*x+c2*x^2+c3*x^3;
```

(%o4) $a1\ x+a0=a1\ x+a0$

(%o5) $b2\ x^2+b1\ x+b0=b2\ x^2+b1\ x+b0$

(%o6) $c3\ x^3+c2\ x^2+c1\ x+c0=c3\ x^3+c2\ x^2+c1\ x+$

Obsérvese que la respuesta (salida) no corresponde a la ordenación de la entrada. Las operaciones suma y producto se pueden llevar a cabo en esta forma general. En el caso del producto, wxMaxima lo escribe como tal sin desarrollar y si se quiere efectuar el producto y expresar-lo como un polinomio en la forma estándar, hace falta explicitarlo con la instrucción “expand” o bien con la instrucción “ratsimp” que expresa el polinomio en la forma estándar.

Así, en el caso de los polinomios anteriores se tendría en el caso de la suma:

```
(%i7) P1 + P2;        P1+P3;
```

(%o7) $b2\ x^2+b1\ x+a1\ x+b0+a0$

(%o8) $c3\ x^3+c2\ x^2+c1\ x+a1\ x+c0+a0$

```
(%i9) ratsimp(P1 + P2);      ratsimp(P1 + P3);
```

(%o9) $b2\ x^2+(b1+a1)\ x+b0+a0$

(%o10) $c3\ x^3+c2\ x^2+(c1+a1)\ x+c0+a0$

En el caso del producto se cumple:

(%i11) P1*P2;

(%o11) $(a1\,x+a0)(b2\,x^2+b1\,x+b0)$

(%i12) P1*P3;

(%o12) $(a1\,x+a0)(c3\,x^3+c2\,x^2+c1\,x+c0)$

(%i13) ratsimp(P1*P2);

(%o13) $a1\,b2\,x^3+(a0\,b2+a1\,b1)\,x^2+(a0\,b1+a1\,b0)\,x+a0\,b0$

(%i14) ratsimp(P1*P3);

(%o14) $a1\,c3\,x^4+(a0\,c3+a1\,c2)\,x^3+(a0\,c2+a1\,c1)\,x^2+(a0\,c1+a1\,c0)\,x+a0\,c0$

-

Obsérvese que la instrucción “expand” no produce ningún efecto adicional en la suma, que se da como resultado en la forma estándar, mientras que si afecta al producto y lo desarrolla tal y como se ha indicado.

También se pueden definir polinomios con raíces conocidas expresadas en forma de parámetros, escritos en forma factorial; por ejemplo:

(%i15) P4:C*(x-a1)*(x-a2)^2;

(%o15) $(x-a1)(x-a2)^2\,C$

(%i16) ratsimp(P4);

(%o16) $(x^3+(-2\,a2-a1)\,x^2+(a2^2+2\,a1\,a2)\,x-a1\,a2^2)\,C$

A continuación se ilustra la sintaxis de la definición y las operaciones con polinomios con ejemplos concretos. Las operaciones suma y diferencia de polinomios se llevan a cabo de forma sencilla teniendo en cuenta la asignación realizada:

(%i1) P1:x^2-4;
P2:x^3-x^2-x+1;

(%o1) x^2-4

(%o2) x^3-x^2-x+1

(%i3) P1 + P2;   P1 - P2;

(%o3) x^3-x-3

(%o4) $-x^3+2\,x^2+x-5$

(%i5) P1*P2;

(%o5) $(x^2-4)(x^3-x^2-x+1)$

(%i6) expand(P1*P2);

(%o6) $x^5-x^4-5\,x^3+5\,x^2+4\,x-4$

Se pueden llevar a cabo combinaciones lineales de polinomios, el resultado de las cuales se obtiene en primera instancia en forma simbólica y si se quiere en forma desarrollada hay que usar, tal como se ha indicado, la instrucción "expand":

```
(%i7) 3*P1-5*P2;
```
(%o7) $3\,(x^2-4)-5\,(x^3-x^2-x+1)$

```
(%i8) expand(3*P1-5*P2);
```
(%o8) $-5\,x^3+8\,x^2+5\,x-17$

Una aplicación útil e importante es la descomposición factorial de un polinomio, que se obtiene directamente con la instrucción "factor" (análoga en la usada en el caso de los números naturales para obtener la descomposición en factores):

```
(%i9) P3:x^6-1;
```
(%o9) x^6-1

```
(%i10) factor(P3);
```
(%o10) $(x-1)(x+1)(x^2-x+1)(x^2+x+1)$

Se pueden calcular con la instrucción "allroots" las raíces de un polinomio:

```
(%i11) allroots(P1);     allroots(P2);
```
(%o11) $[x=2.0,x=-2.0]$

(%o12) $[x=0.99999998509885,x=1.000000014901168,x=-1.00000000000001$

En el segundo caso, los algoritmos numéricos del programa producen un resultado que puede parecer sorprendente; la interpretación de estos resultados permite concluir que las raíces son 1 (doble) y –1 (simple), aspecto que podemos comprobar fácilmente con la descomposición factorial del polinomio P2:

```
(%i13) factor(P2);
```
(%o13) $(x-1)^2\,(x+1)$

Se puede calcular de forma directa el cociente y el residuo de la división de dos polinomios con la instrucción "divide":

```
(%i14) divide(P2,P1);
       divide(x^4-x^3+x-1, x+1);
       divide(x^4-x^3+x-1, x+2);
```
(%o14) $[x-1,3\,x-3]$

(%o15) $[x^3-2\,x^2+2\,x-1,0]$

(%o16) $[x^3-3\,x^2+6\,x-11,21]$

También se puede obtener directamente el máximo común divisor y el mínimo común múltiple de dos polinomios; en este segundo caso hace falta cargar previamente la librería "functs":

```
(%i17) gcd(x^2-2*x+1, x^2-1);
```

(%o17) $x-1$

```
(%i18) load(functs)$
define: warning: redefining the built-in function lcm
(%i19) lcm(x^2-2*x, x^2-3*x);
```

(%o19) $(x-3)(x-2)x$

Las fracciones racionales se definen de forma sencilla como cociente entre dos polinomios; así mismo, se puede obtener la descomposición de una fracción racional en elementos o fracciones simples con la instrucción "partfrac":

```
(%i1) P1:x^2-4;
      P2:x^3-x^2-x+1;
```

(%o1) x^2-4

(%o2) x^3-x^2-x+1

```
(%i3) P1/P2;
```

(%o3) $\dfrac{x^2-4}{x^3-x^2-x+1}$

```
(%i4) P1/P2 = partfrac(P1/P2, x);
```

(%o4) $\dfrac{x^2-4}{x^3-x^2-x+1}=-\dfrac{3}{4(x+1)}+\dfrac{7}{4(x-1)}-\dfrac{3}{2(x-1)^2}$

Se pueden definir primero los polinomios y después la fracción racional con el cociente de los polinomios considerados:

```
(%i5) Q1:x^2-2*x+4;
      Q2:x^8+5*x^7+13*x^6+22*x^5+26*x^4+22*x^3+13*x^2+5*x+1;
```

(%o5) x^2-2x+4

(%o6) $x^8+5x^7+13x^6+22x^5+26x^4+22x^3+13x^2+5x+1$

```
(%i7) '(Q1/Q2) = partfrac(Q1/Q2, x);
```

(%o7) $\dfrac{Q1}{Q2}=\dfrac{-17x-7}{x^2+x+1}+\dfrac{-10x-7}{(x^2+x+1)^2}+\dfrac{-3x-6}{(x^2+x+1)^3}+\dfrac{17}{x+1}+\dfrac{7}{(x+1)^2}$

La descomposición de una fracción racional en fracciones simples se puede llevar a cabo con polinomios que contengan parámetros. Por ejemplo:

```
(%i8) Q3:x+1;
      Q4:(x-a)*(x^2-b^2);
```

(%o8) $x+1$

(%o9) $(x-a)(x^2-b^2)$

```
(%i10) Q3/Q4 = partfrac(Q3/Q4, x);
```

(%o10) $\frac{x+1}{(x-a)(x^2-b^2)} = -\frac{b-1}{(2b^2+2ab)(x+b)} + \frac{b+1}{(2b^2-2ab)(x-b)} - \frac{a+1}{(b^2-a^2)(x-a)}$

02-4.- Resolución básica de ecuaciones

Los cálculos de este apartado corresponden al fichero **Tema_02-4.wxm**.

Las ecuaciones se pueden introducir con la metodología vista anteriormente de la sintaxis de variables; esto permite recuperarlas si hace falta más adelante y no es necesario introducirlas nuevamente. Ejemplos de sintaxis e introducción de ecuaciones:

```
(%i1) Eq01:2*x+1=x-4;
```
(%o1) $2\,x+1=x-4$

```
(%i2) Eq02:a*x+b=c;
```
(%o2) $a\,x+b=c$

```
(%i3) Eq03:4*x^2=5-x;
```
(%o3) $4\,x^2=5-x$

```
(%i4) Eq04:x^3-5*x^2-17*x+21=0;
```
(%o4) $x^3-5\,x^2-17\,x+21=0$

```
(%i5) Eq05:x^4-5*x^3-43*x^2+53*x+90=0;
```
(%o5) $x^4-5\,x^3-43\,x^2+53\,x+90=0$

Se pueden explicitar los dos miembros de una ecuación introducida y definir una nueva ecuación a partir de los miembros de otras ecuaciones:

```
(%i6) first(Eq03);
      second(Eq03);
      Eq06:second(Eq03)=first(Eq04);
```
(%o6) $4\,x^2$

(%o7) $5-x$

(%o8) $5-x=x^3-5\,x^2-17\,x+21$

También se puede definir una nueva ecuación a partir de operaciones con otras ecuaciones:

```
(%i9) Eq07:Eq03+Eq04;
```
(%o9) $x^3-x^2-17\,x+21=5-x$

```
(%i10) Eq08:2*Eq03-4*Eq04;
```
(%o10) $8\,x^2-4\,(x^3-5\,x^2-17\,x+21)=2\,(5-x)$

```
(%i11) expand(%);
```
(%o11) $-4\,x^3+28\,x^2+68\,x-84=10-2\,x$

Para la resolución de ecuaciones hace falta usar la instrucción “solve” indicando la referencia asignada a la ecuación y la variable usada en la ecuación:

```
(%i12) solve(Eq01, x);
(%o12) [x=-5]
(%i13) solve(Eq02, x);
```

(%o13) $[x=\frac{c-b}{a}]$

```
(%i14) solve(Eq03, x);
```

(%o14) $[x=-\frac{5}{4},x=1]$

```
(%i15) solve(Eq04, x);
(%o15) [x=1,x=-3,x=7]
(%i16) solve(Eq05, x);
(%o16) [x=9,x=-1,x=2,x=-5]
(%i17) solve(Eq07, x);
(%o17) [x=-4,x=4,x=1]
```

Por defecto el programa da las soluciones de una ecuación en forma “exacta” o simbólica, si esto es posible. Así, por ejemplo, las soluciones que se obtienen de las ecuaciones Eq08 i Eq09 son:

```
(%i18) solve(Eq08, x);
```

(%o18) $[x=-\frac{\sqrt{130}-6}{2},x=\frac{\sqrt{130}+6}{2},x=1]$

```
(%i19) Eq09:1.44*x^3-7.488*x^2-4.59*x+46.575=0;
```

(%o19) $1.44\,x^3-7.488\,x^2-4.59\,x+46.575=0$

```
(%i20) solve(Eq09, x);
rat: replaced 46.575 by 1863/40 = 46.575
  rat: replaced -4.59 by -459/100 = -4.59
  rat: replaced -7.488 by -936/125 = -7.488
  rat: replaced 1.44 by 36/25 = 1.44
```

(%o20) $[x=-\frac{23}{10},x=\frac{15}{4}]$

Para obtener las soluciones en forma numérica hay que aplicar la instrucción “numer”. Por ejemplo, en el mismo caso de les ecuaciones Eq08 i Eq09 se obtienen las soluciones:

```
(%i21) solve(Eq08, x), numer;
rat: replaced 22.80350850198276 by 6499/285 = 22.80350877192982
  rat: replaced 22.80350850198276 by 6499/285 = 22.803508771929
   (%o21) [x=-2.700877192982456,x=8.700877192982457,x=1]
```

```
(%i22) solve(Eq09, x), numer;
rat: replaced 46.575 by 1863/40 = 46.575
   rat: replaced -4.59 by -459/100 = -4.59
   rat: replaced -7.488 by -936/125 = -7.4{
   rat: replaced 1.44 by 36/25 = 1.44
   rat: replaced 46.575 by 1863/40 = 46.5:
   rat: replaced -4.59 by -459/100 = -4.59
   rat: replaced -7.488 by -936/125 = -7.4{
   rat: replaced 1.44 by 36/25 = 1.44
   rat: replaced 0.009 by 9/1000 = 0.009
(%o22) [x=-2.3,x=3.75]
```

Más adelante se verán otros métodos para resolver ecuaciones, basados en la Teoría de funciones continuas. Así mismo se estudiarán otros tipos de ecuaciones no tan sencillas como les estudiadas en esta sección.

02-5.- Resolución básica de sistemas de ecuaciones

Los cálculos de este apartado corresponden al fichero **Tema_02-5.wxm**.

La sintaxis de definición de los sistemas de ecuaciones es sencilla y consiste en asignar una denominación a cada una de las ecuaciones del sistema y finalmente una etiqueta al sistema. El programa puede trabajar con sistemas que contienen parámetros. Por ejemplo:

```
(%i1) Eq001: a*x+y=a;
      Eq002:x+a*y=a;
(%o1) y+a x=a
(%o2) a y+x=a
(%i3) Sist01:[Eq001,Eq002];
(%o3) [y+a x=a,a y+x=a]
```

Se puede organizar toda la información en una sola línea de entrada; por ejemplo:

```
(%i4) Eq003:x+2*y=0;      Eq004:3*x+5*y=6;
      Sist02:[Eq003,Eq004];
(%o4) 2 y+x=0
(%o5) 5 y+3 x=6
(%o6) [2 y+x=0,5 y+3 x=6]
```

Se pueden considerar y resolver sistemas con un mayor número de ecuaciones; por ejemplo si el sistema tiene tres ecuaciones:

```
(%i7) Eq010:x+y-z=0$      Eq011:x-2*y+z=0$      Eq012:2*x-y+2*z-6=0$
      Sist03:[Eq010,Eq011,Eq012];
(%o10) [-z+y+x=0,z-2 y+x=0,2 z-y+2 x-6=0]
```

La resolución se lleva a cabo con la misma instrucción o usando las opciones del menú Ecuaciones, “Resolver sistema lineal” cuando se trata de sistemas de ecuaciones lineales:

```
(%i11) solve(Sist01, [x,y]);
```

(%o11) $[[x=\frac{a}{a+1},y=\frac{a}{a+1}]]$

```
(%i12) solve(Sist02, [x,y]);
(%o12) [[x=12,y=-6]]
(%i13) solve(Sist03, [x,y,z]);
(%o13) [[x=1,y=2,z=3]]
```

Se pueden resolver también sistemas de ecuaciones no lineales, aplicando la instrucción “algsys” o bien en el menú Ecuaciones, “Resolver sistema algebraico”:

(%i14) Eq020:x^4-y^2=1; Eq021:x^2+y=2;
Sist04:[Eq020,Eq021];

(%o14) $x^4 - y^2 = 1$

(%o15) $y + x^2 = 2$

(%o16) $[x^4 - y^2 = 1, y + x^2 = 2]$

(%i17) algsys([Eq020, Eq021], [x,y]);

(%o17) $[[x = -\frac{\sqrt{5}}{2}, y = \frac{3}{4}], [x = \frac{\sqrt{5}}{2}, y = \frac{3}{4}]]$

(%i18) Eq022:3*x^2-y^2=6; Eq023:x-y=0;
Sist05:[Eq022,Eq023];

(%o18) $3\,x^2 - y^2 = 6$

(%o19) $x - y = 0$

(%o20) $[3\,x^2 - y^2 = 6, x - y = 0]$

(%i21) algsys([Eq022, Eq023], [x,y]);

(%o21) $[[x = -\sqrt{3}, y = -\sqrt{3}], [x = \sqrt{3}, y = \sqrt{3}]]$

02-6.- Resolución básica de inecuaciones

Los cálculos de este apartado corresponden al fichero **Tema_02-6.wxm**.

Las inecuaciones tienen una sintaxis sencilla, similar a la de las ecuaciones. Supongamos, por ejemplo, que se quiere resolver la inecuación $\frac{1}{x-2}<1$; la sintaxis en wxMaxima es:

```
(%i1) Ineq01:1/(x-2)<1;
```

(%o1) $\frac{1}{x-2}<1$

Ara observamos que la inecuación se puede transformar de la manera siguiente:

$$\frac{1}{x-2}<1 \quad \Leftrightarrow \quad \frac{1}{x-2}-1<0 \quad \Leftrightarrow \quad \frac{1-(x-2)}{x-1}=\frac{3-x}{x-2}<0$$

Por lo tanto, se escribirá:

```
(%i2) Ineq01b:(3-x)/(x-2)<0;
```

(%o2) $\frac{3-x}{x-2}<0$

La resolución de una inecuación se lleva a cabo con la introducción de una condición o condiciones sobre el dominio de la variable y la verificación del signo de la condición correspondiente a la inecuación; así en el caso anterior hay que considerar los casos $x<2$, $2<x<3$ i $x>3$.

La sintaxis es la siguiente:

```
(%i3) assume(x<2);
      is((3-x)/(x-2)<0);
```

(%o3) $[x<2]$

(%o4) true

Así pues, la inecuación es cierta en este caso; ahora hay que introducir la nueva condición pero también hay que eliminar la condición introducida anteriormente:

```
(%i5) forget(x<2)$
      assume(x>2, x<3);
      is((3-x)/(x-2)<0);
```

(%o6) $[x>2,x<3]$

(%o7) false

Continuando con la metodología, eliminaremos la última condición introducida y pondremos una nueva, para acabar de resolver la inecuación:

```
(%i8) forget(x>2, x<3)$
      assume(x>3);
      is((3-x)/(x-2)<0);
(%o9) [x>3]
(%o10) true
```

Finalmente, pues, se puede concluir que la inecuación es cierta en el intervalo $]-\infty, 2[\cup]3, +\infty[$.

02-7.- Funciones elementales con wxMaxima

Los cálculos de este apartado corresponden al fichero **Tema_02-7.wxm**.

En la Tabla 02-1 se puede ver la descripción de las funciones elementales y su notación o sintaxis en wxMaxima. Es conveniente saber esta notación para aplicar correctamente los cálculos que involucren las funciones mencionadas.

Función	**Notación matemática usual**	**Notación wxMaxima**
Valor absoluto de x	$\|x\|$	abs(x)
Raíz cuadrada de x	$\sqrt{x}$	sqrt(x)
Exponencial real	exp(x)	exp(x)
Logaritmo neperiano	ln(x) o bien log(x)	log(x)
Seno	sin(x)	sin(x)
Cosenos	cos(x)	cos(x)
Tangente	tan(x)	tan(x)
Arco-seno	arcsin(x)	asin(x)
Arco-coseno	arccos(x)	acos(x)
Arco-tangente	arctan(x)	atan(x)
Seno hiperbólico	sinh(x)	sinh(x)
Coseno hiperbólico	cosh(x)	cosh(x)
Tangente hiperbólica	tanh(x)	tanh(x)
Argumento seno hiperbólico	asinh(x)	asinh(x)
Argumento coseno hiperbólico	acosh(x)	acosh(x)
Argumento tangente hiperbólica	atanh(x)	atanh(x)

Tabla 02-1 Funciones elementales predefinidas en wxMaxima: denominación, notación habitual y notación de wxMaxima.

Para calcular el valor de una función predefinida en un punto sólo hace falta escribir la notación de la función y el punto a continuación entre paréntesis. Así, por ejemplo, la sintaxis para trabajar con las funciones “valor absoluto”, “raíz cuadrada” i “exponencial” es:

```
(%i1) abs((-2.35)^3-4*(7.25^2));
(%o1) 223.227875
(%i2) sqrt(8975.3456);   sqrt(10);   sqrt(10.0);
(%o2) 94.73830059696026
(%o3) √10
(%o4) 3.16227766016838
```

```
(%i5) exp(0);     exp(1);     exp(1.0);      exp(10.45);
(%o5) 1
  (%o6) %e
  (%o7) 2.718281828459045
  (%o8) 34544.37470927775
```

Se observa que la raíz cuadrada de un número natural se da indicando el número debajo del símbolo de la raíz cuadrada; si se quiere el resultado en forma numérica hace falta usar la instrucción “numer” o introducir el número con coma flotante para que sea interpretado como número real.

Veamos ahora el cálculo del valor de la función logaritmo neperiano (o logaritmo natural) en diferentes puntos; hay que hacer notar que en wxMaxima la función logaritmo neperiano se designa “log”, siguiendo la tendencia más moderna en Análisis Matemática y que algunos textos ya van incorporando; la notación clásica, que aún es la más habitual, es “ln”. Algunos ejemplos de cálculos:

```
(%i9) log(0.5);      log(1.0);      log(2.0);   log(%e);
      log(%e^2);   log(%e^n);   log(10);     log(10.0);
(%o9) -0.69314718055995
 (%o10) 0.0
 (%o11) 0.69314718055995
 (%o12) 1
 (%o13) 2
 (%o14) n
 (%o15) log(10)
 (%o16) 2.302585092994046
```

Una cuestión importante es el cambio de base en la función logarítmica, que permite la definición de la función logarítmica de base cualquier número $b>0, b\neq 1$:

```
(%i17) log_b(x):=(log(x)/log(b));    b:12.0$    x:2.0$    log_b(x);
```

$$(\%o17)\ log\_b(x):=\frac{log(x)}{log(b)}$$

```
 (%o20) 0.27894294565113
(%i21) b:12.3450$   x:2.98760$    log_b(x);
(%o23) 0.43547991665977
```

El argumento de las funciones trigonométricas por defecto es en radianes; para calcular el valor de les funciones trigonométricas de un ángulo expresado en grados sexagesimales o centesimales, hay que efectuar la conversión del correspondiente valor en radianes:

$$\text{Angulo (radianes)} = \text{Angulo (grados sexagesimales)} \cdot \frac{\pi}{180}$$

$$\text{Angulo (radianes)} = \text{Angulo (grados centesimales)} \cdot \frac{\pi}{200}$$

Cálculo del seno de diferentes ángulos:

```
(%i1) x1:1.0$     x2:%pi$     x3:3*(%pi)/2$     x4:2*(%pi)$
      sin(x1);    sin(x2);    sin(x3);          sin(x4);
(%o5) 0.8414709848079
(%o6) 0
(%o7) -1
(%o8) 0
```

Cálculo del coseno de diferentes ángulos:

```
(%i9) cos(x1);   cos(x2);   cos(x3);   cos(x4);
(%o9) 0.54030230586814
(%o10) -1
(%o11) 0
(%o12) 1
```

Definición de las funciones seno y el coseno cuando el ángulo se expresa en grados sexagesimales:

```
(%i13) sin_g(x):=sin(x*(%pi)/180);
       cos_g(x):=cos(x*(%pi)/180);
```

$$(\%o13)\ \mathrm{sin\_g}(x):=\sin\left(\frac{x\,\pi}{180}\right)$$

$$(\%o14)\ \mathrm{cos\_g}(x):=\cos\left(\frac{x\,\pi}{180}\right)$$

Ejemplos de cálculos con estas funciones:

```
(%i15) y1:30$
       y2:45$
       y3:60$
       y4:90$
```

(%i19) sin_g(y1);   sin_g(y2);   sin_g(y3);   sin_g(y4);

(%o19) $\frac{1}{2}$

(%o20) $\frac{1}{\sqrt{2}}$

(%o21) $\frac{\sqrt{3}}{2}$

(%o22) 1

(%i23) cos_g(y1);   cos_g(y2);   cos_g(y3);   cos_g(y4);

(%o23) $\frac{\sqrt{3}}{2}$

(%o24) $\frac{1}{\sqrt{2}}$

(%o25) $\frac{1}{2}$

(%o26) 0

Definición del seno y del coseno cuando el ángulo se expresa en grados centesimales:

```
(%i27) sin_c(x):=sin(x*(%pi)/200);
       cos_c(x):=cos(x*(%pi)/200);
```

(%o27) $\mathrm{sin\_c}(x):=\sin\left(\frac{x\,\pi}{200}\right)$

(%o28) $\mathrm{cos\_c}(x):=\cos\left(\frac{x\,\pi}{200}\right)$

Ejemplos de cálculos con estas funciones:

```
(%i29) y1:25$      y2:50$      y3:100$
       sin_c(y1), numer;  sin_c(y2), numer;  sin_c(y3), numer;
```

(%o32) 0.38268343236509

(%o33) 0.70710678118655

(%o34) 1.0

```
(%i35) y1:25$      y2:50$      y3:100$
       cos_c(y1),numer;  cos_c(y2),numer;  cos_c(y3),numer;
```

(%o38) 0.92387953251129

(%o39) 0.70710678118655

(%o40) $6.1230317691118863\ 10^{-17}$

Cálculo del valor de la función tangente en diferentes puntos:

```
(%i41) tan(%pi/4);   tan(%pi);    tan(3*%pi/4);    tan(5*%pi/14);
(%o41) 1
  (%o42) 0
  (%o43) -1
```

(%o44) $\tan\left(\frac{5\pi}{14}\right)$

```
(%i45) tan(5*%pi/14), numer;
(%o45) 2.076521396572336
```

Cálculo del valor de la función arco-seno en diferentes puntos:

```
(%i46) asin(0);   asin(0.5);    asin(1);
(%o46) 0
  (%o47) 0.5235987755983
```

(%o48) $\frac{\pi}{2}$

Cálculo del valor de la función arco-coseno en diferentes puntos:

```
(%i49) acos(0);   acos(0.5);    acos(1);
```

(%o49) $\frac{\pi}{2}$

```
  (%o50) 1.047197551196598
  (%o51) 0
```

Cálculo del valor de la función arco-tangente en diferentes puntos:

```
(%i52) atan(0);   atan(2.0);     atan(5.0);
(%o52) 0
  (%o53) 1.10714871779409
  (%o54) 1.373400766945016
```

Cálculo del valor de la función seno hiperbólico en diferentes puntos:

```
(%i1) sinh(0);      sinh(2.0);    sinh(5.0);
(%o1) 0
  (%o2) 3.626860407847019
  (%o3) 74.20321057778875
```

Cálculo del valor de la función coseno hiperbólico en diferentes puntos:

```
(%i4) cosh(0);      cosh(2.0);      cosh(5.0);
(%o4) 1
(%o5) 3.762195691083631
(%o6) 74.20994852478785
```

Cálculo del valor de la función tangente hiperbólica en diferentes puntos:

```
(%i7) tanh(0);      tanh(2.0);      tanh(5.0);
(%o7) 0
(%o8) 0.96402758007582
(%o9) 0.9999092042626
```

La sintaxis para la definición de una función ya se ha comentado en el apartado anterior. El estudio más exhaustivo de las funciones se verá en sesiones posteriores. A continuación se muestra un ejemplo de función definida mediante operaciones con funciones elementales y el cálculo de su valor en tres puntos de su campo de existencia.

```
(%i1) f(x):=2*sin(x)-3*exp(x)+log(abs((x-1)/(x-2)));
```

(%o1) $f(x):=2\sin(x)-3\exp(x)+\log\left(\left|\frac{x-1}{x-2}\right|\right)$

```
(%i2) f(3.45);  f(-2.54);  f(2.0001);
(%o2) -94.58373548353652
(%o3) -1.617311945177004
(%o4) -11.14043314229555
```